Chapter 6
Using the Output Delivery System

Chapter Table of Contents

OVERVIEW .......................................................... 181
Output Objects and ODS Destinations .................................. 181
Using the Output Delivery System ...................................... 182
Compatibility Issues with Version 6 Prototypes ...................... 188

EXAMPLES .......................................................... 189
Example 6.1 Creating HTML Output with ODS ...................... 189
Example 6.2 Creating HTML Output with a Table of Contents ...... 191
Example 6.3 Determining the Names of ODS Tables .................. 193
Example 6.4 Selecting ODS Tables for Display ...................... 195
Example 6.5 Creating an Output Data Set from an ODS Table ....... 199
Part 1. General Information
Chapter 6
Using the Output Delivery System

Overview

In the latest version of SAS software, all SAS/ETS procedures use the Output Delivery System (ODS) to manage their output. This includes managing the form in which the output appears as well as its organization and format. The default for SAS/ETS procedures is to produce the usual SAS listing file. However, by using the features of the Output Delivery System, you can make changes to the format and appearance of your SAS output. In particular, you can

- display your output in hypertext markup language (HTML).
- display your output in Rich-Text-Format (RTF).
- create SAS data sets directly from output tables.
- select or exclude individual output tables.
- customize the layout, format, and headers of your output.

ODS features can provide you with a powerful tool for managing your output. This chapter provides background material and illustrates typical applications of ODS with SAS/ETS software.

For complete documentation on the Output Delivery System, refer to The Complete Guide to the SAS Output Delivery System.

Output Objects and ODS Destinations

All SAS procedures produce output objects that the Output Delivery System delivers to various ODS destinations, according to the default specifications for the procedure or to your own specifications.

All output objects (for example, a table of parameter estimates) consist of two component parts:

- the data component, which consists of the results computed by a SAS procedure.
- the template, which contains rules for formatting and displaying the results.

When you invoke a SAS procedure, the procedure sends all output to the Output Delivery System. ODS then routes the output to all open destinations. You define

*experimental in Version 7
the form the output should take when you specify an ODS destination. Supported destinations are as follows:

- Listing destination (the standard SAS listing), which is the default.
- HTML destination, hypertext markup language.
- Output destination, SAS data set.

Future versions of ODS will support the following additional destinations:

- the ODS Output Document for modifying and replaying output without rerunning the procedure that created it.
- Rich Text Format (RTF) for inclusion in Microsoft Word.
- postscript and PCL for high fidelity printers.

You can activate multiple ODS destinations at the same time, so that a single procedure step can route output to multiple destinations. If you do not supply any ODS statements, ODS delivers all output to the SAS listing, which is the default.

Each output object has an associated template that defines its presentation format. You can modify the presentation of the output by using the TEMPLATE procedure to alter these templates or to create new templates. You can also specify stylistic elements for ODS destinations, such as cell formats and headers, column ordering, colors, and fonts. For detailed information, refer to the chapter titled “The Template Procedure” in the SAS Procedures Guide.

### Using the Output Delivery System

The ODS statement is a global statement that enables you to provide instructions to the Output Delivery System. You can use ODS statements to specify options for different ODS destinations, select templates to format your output, and select and exclude output. You can also display the names of individual output tables as they are generated.

In order to select, exclude, or modify a table, you must first know its name. You can obtain the table names in several ways:

- For any SAS/ETS procedure, you can obtain table names from the individual procedure chapter or from the individual procedure section of the SAS online Help system.
- For any SAS procedure, you can use the SAS Explorer window to view the names of the tables created in your SAS run (see the section “Using ODS with the SAS Explorer” on page 185 for more information).
- For any SAS procedure, you can use the ODS TRACE statement to find the names of tables created in your SAS run. The ODS TRACE statement writes identifying information to the SAS log (or, optionally, to the SAS listing) for each generated output table.
Specify the ODS TRACE ON statement prior to the procedure statements that create the output for which you want information. For example, the following statements write the trace record for the specific tables created in this AUTOREG procedure step.

```sas
ods trace on;
proc autoreg;
  model y1 = time;
  model y2 = time;
run;
```

By default, the trace record is written to the SAS log, as displayed in Figure 6.1. Alternatively, you can specify the LISTING option, which writes the information, interleaved with the procedure output, to the SAS listing (see Example 6.1).

```
ods trace on;
proc autoreg;
  model y1 = time;
  model y2 = time;
run;
.
.
.
```

Output Added:
--------------
Name: ParameterEstimates
Label: Parameter Estimates
Template: ets.autoreg.ParameterEstimates
Path: Autoreg.Model1.OLSEst.ParameterEstimates
--------------
.
.
.
```
```
```
```
Output Added:
--------------
Name: ParameterEstimates
Label: Parameter Estimates
Template: ets.autoreg.ParameterEstimates
Path: Autoreg.Model2.OLSEst.ParameterEstimates
--------------
.
.
.
```

**Figure 6.1.** Partial Contents of the SAS Log: Result of the ODS TRACE Statement

Figure 6.1 displays the trace record, which contains the name of each created table and its associated label, template, and path. The label provides a description of the table. The template name displays the name of the template used to format the table. The path shows the output hierarchy to which the table belongs.

The fully qualified path is given in the trace record. A partially qualified path consists of any part of the full path that begins immediately after a period (.) and continues to the end of the full path. For example, the full path for the parameter estimates for the first model in the preceding regression analysis is

```
Autoreg.Modell.OLSEst.ParameterEstimates
```
Therefore, partially qualified paths for the table are

\[
\text{Autoreg.Model1.OLSEst.ParameterEstimates} \\
\text{Model1.OLSEst.ParameterEstimates} \\
\text{OLSEst.ParameterEstimates} \\
\text{ParameterEstimates}
\]

To refer to a table (in order to select or exclude it from display, for example), specify either the table name or use the table's fully or partially qualified path. You may want to use qualified paths when your SAS program creates several tables that have the same name, as in the preceding example. In such a case, you can use a partially qualified path to select a subset of tables, or you can use a fully qualified path to select a particular table.

You specify the tables that ODS selects or excludes with the ODS SELECT or ODS EXCLUDE statement. Suppose that you want to display only the tables of parameter estimates from the preceding regression analysis. You can give any of the following statements (before invoking the AUTOREG procedure) to display both tables of parameter estimates. For this example, these statements are equivalent:

\[
\text{ods select Autoreg.Model1.OLSEst.ParameterEstimates} \\
\text{Autoreg.Model2.OLSEst.ParameterEstimates;}
\]

\[
\text{ods select Model1.OLSEst.ParameterEstimates} \\
\text{Model2.OLSEst.ParameterEstimates;}
\]

\[
\text{ods select OLSEst.ParameterEstimates;}
\]

\[
\text{ods select ParameterEstimates;}
\]

The first ODS SELECT statement specifies the full path for both tables. The second statement specifies the partially qualified path for both tables. The third and fourth statements specify the partial path “OLSEst.ParameterEstimates,” and single name “ParameterEstimates,” which are shared by both tables.

The Output Delivery System records the specified table names in its internal selection or exclusion list. ODS then processes the output it receives. Note that ODS maintains an overall selection or exclusion list that pertains to all ODS destinations, and it maintains a separate selection or exclusion list for each ODS destination. The list for a specific destination provides the primary filtering step. Restrictions you specify in the overall list are added to the destination-specific lists.

Suppose, for example, that your listing exclusion list (that is, the list of tables you wish to exclude from the SAS listing) contains the “Summary” table, which you specify with the statement

\[
\text{ods listing exclude Summary;}
\]

and your overall selection list (that is, the list of tables you want to select for all destinations) contains the tables “Summary” and “ParameterEstimates,” which you specify with the statement
ods select ParameterEstimates Summary;

The Output Delivery System then sends only the “ParameterEstimates” and “Summary” tables to all open destinations except the SAS listing. It sends only the “ParameterEstimates” table to the SAS listing because the table “Summary” is excluded from that destination.

Some SAS procedures, such as the ARIMA or the MODEL procedure, support run-group processing, which means that a RUN statement does not end the procedure. A QUIT statement explicitly ends such procedures; if you omit the QUIT statement, a PROC or a DATA statement implicitly ends such procedures. When you use the Output Delivery System with procedures that support run-group processing, it is good programming practice to specify a QUIT statement at the end of the procedure. This causes ODS to clear the selection or exclusion list, and you are less likely to encounter unexpected results.

Using ODS with the SAS Explorer

The SAS Explorer is a new feature that enables you to examine the various parts of the SAS System. Figure 6.2 displays the Results window from the SAS Explorer. The Results node retains a running record of your output as it is generated during your SAS session. Figure 6.2 displays the output hierarchy when the preceding statements are executed.

![Figure 6.2. The Results Window from the SAS Explorer](image)

When you click on the output table names in the Results window, you link directly to the output in the output window or, if you specify the HTML destination, in an HTML browser. The items on the left-hand side of the Results node are output directories. The items on the right-hand side of the Results node are the names of the actual output objects. You can also use the Explorer to determine names of the templates
associated with each output table.

**Controlling Output Appearance with Templates**

A template is an abstract description of how output should appear when it is formatted. Templates describe several characteristics of the output, including headers, column ordering, style information, justification, and formats. All SAS/ETS procedures have templates, which are stored in the SASHELP library.

You can create or modify a template with the TEMPLATE procedure. For example, you can specify different column headings or different orderings of columns in a table. You can find the template associated with a particular output table by using the ODS TRACE statement or the SAS Explorer.

You can display the contents of a template by executing the following statements:

```sas
proc template;
   source templatename;
run;
```

where `templatename` is the name of the template.

Suppose you want to change the way all of the parameter estimates are displayed by the AUTOREG procedure. You can redefine the templates that the procedure uses with PROC TEMPLATE. For example, in order to have the ESTIMATE and STANDARD ERROR columns always displayed with more digits, you can redefine the columns used by the procedure to display them:

```sas
proc template;
   edit ets.autoreg.ParameterEstimates;
      edit Estimate; format=Best16.; end;
      edit StdErr; format=Best16.; end;
   end;
run;
```

The BESTw. format enables you to display the most information about a value, according to the available field width. The BEST16. format specifies a field width of 16. Refer to the chapter on formats in *SAS Language Reference: Dictionary* for detailed information.

When you run PROC TEMPLATE to modify or edit a template, the template is stored in your SASUSER library. You can then modify the path that ODS uses to look up templates with the ODS PATH statement in order to access these new templates in a later SAS session. This means that you can create a default set of templates to modify the presentation format for all your SAS output. (Note that you can specify the SHOW option in the ODS PATH statement to determine the current path.)

It is important to note the difference between a style template and a table template. A table template applies only to the specific tables that reference the template. The preceding statements that modify the “ets.autoreg.ParameterEstimates” template provide an example of modifying columns within a table template.

A style template applies to an entire SAS job and can be specified only in the ODS
HTML statement. You can specify a style as follows:

    ods html style=Styles.Brown;

A style template controls stylistic elements such as colors, fonts, and presentation attributes. When you use a style template, you ensure that all your output shares a consistent presentation style.

You can also reference style information in table templates for individual headers and data cells. You can modify either type of template with the TEMPLATE procedure. For information on creating your own styles, refer to The Complete Guide to the SAS Output Delivery System.

**Interaction Between ODS and the NOPRINT Option**

Most SAS/ETS procedures support a NOPRINT option that you can use when you want to create an output data set but do not want any displayed output. Typically, you use an OUTPUT statement in addition to the procedure’s NOPRINT option to create a data set and suppress displayed output.

You can also use the Output Delivery System to create output data sets by using the ODS OUTPUT statement. However, if you specify the NOPRINT option, the procedure may not send any output to the Output Delivery System. Therefore, when you want to create output data sets through ODS (using the ODS OUTPUT statement), and you want to suppress the display of all output, specify

    ODS SELECT NONE;

or close the active ODS destinations by giving the command

    ODS destinationname CLOSE;

where destinationname is the name of the active ODS destination (for example, ODS HTML CLOSE).

**Note:** The ODS statement does not instruct a procedure to generate output: instead, it specifies how the Output Delivery System should manage the table once it is created. The requested data table (output) has to be generated by the procedure before ODS can manage it. You must ensure that the proper options are in effect. For example, the following code does not create the requested data set Parms.

    proc autoreg;
        ods output ML.ParameterEstimates=Parms;
        model y1 = time;
    run;

When you execute these statements, the following line is displayed in the log:

    WARNING: Output 'ML.ParameterEstimates' was not created.
The data set `Parms` is not created because the table of parameter estimates is generated only when the METHOD=ML option is specified in the MODEL statement in the AUTOREG procedure.

### Compatibility Issues with Version 6 Prototypes

- The Version 6 prototype of the ODS output hierarchy is stored in a SAS catalog. The latest version of SAS software has a more flexible item-store file type used to store templates and ODS output.

- The Version 6 prototype ODS uses two macro variables (\_DISK\_ and \_PRINT\_) to regulate the saving of an output hierarchy. The latest version of SAS software uses the global ODS statement to accomplish this task.

- The Version 6 PROC TEMPLATE and PROC OUTPUT syntax is not compatible with the latest version of SAS software.
Examples

The following examples display typical uses of the Output Delivery System.

Example 6.1. Creating HTML Output with ODS

This example demonstrates how you can use the ODS HTML statement to display your output in hypertext markup language (HTML).

The following statements create the data set AR2, which contains a second-order autocorrelated time series \( Y \). The AUTOREG procedure is then invoked to estimate the time trend of \( Y \).

The ODS HTML statement specifies the name of the file to contain body of the HTML output.

```sas
data AR2;
  ul = 0; ull = 0;
  do Time = -10 to 36;
    u = + 1.3 * ul - .5 * ull + 2*rannor(12346);
    Y = 10 + .5 * time + u;
    if Time > 0 then output;
    ull = ul; ul = u;
  end;
run;
ods html body='trend.htm';

ods html close;

proc autoreg;
  model Y = Time;
run;
ods html close;
```

By default, the SAS listing receives all output generated during your SAS run. In this example, the ODS HTML statement opens the HTML destination, and both destinations receive the generated output. Output 6.1.1 displays the results as they are displayed in the SAS listing.

Note that you must specify the following statement before you can view your output in a browser.

```sas
ods html close;
```

If you do not close the HTML destination, your HTML file may contain no output, or you may experience other unexpected results.

Output 6.1.2 displays the file 'trend.htm’, which is specified in the preceding ODS HTML statement.
Part 1. General Information

Output 6.1.1. Results for PROC AUTOREG: SAS Listing Output

Estimated Time Trend of Y
The AUTOREG Procedure
Dependent Variable  Y

Ordinary Least Squares Estimates

| Variable          | DF | Estimate | Standard Error | t Value | Pr > |t| |
|-------------------|----|----------|----------------|---------|------|---|
| Intercept         | 1  | 8.2308   | 0.8559         | 9.62    | <.0001 |
| Time              | 1  | 0.5021   | 0.0403         | 12.45   | <.0001 |

Output 6.1.2. Results for PROC AUTOREG: HTML Output
Example 6.2. Creating HTML Output with a Table of Contents

The following example uses ODS to display the output in HTML with a table of contents.

The data is the population of the United States in millions recorded at ten year intervals starting in 1790 and ending in 1990. The MODEL procedure is used to estimate a logistic growth curve by nonlinear ordinary least squares.

```sas
data uspop;
  input pop :6.3 @@;
  retain year 1780;
  year=year+10;
  label pop='U.S. Population in Millions';
  datalines;
  3929 5308 7239 9638 12866 17069 23191 31443
  39818 50155 62947 75994 91972 105710 122775 131669
  151325 179323 203211 226542 248710
;
ods html body='uspop.htm'
  contents='uspopc.htm'
  frame='uspopf.htm';
  title 'Logistic Growth Curve Model of U.S. Population';
  proc model data=uspop;
    label a = 'Maximum Population'
      b = 'Location Parameter'
      c = 'Initial Growth Rate';
    pop = a / ( 1 + exp( b - c * (year-1790) ) );
    fit pop start=(a 1000 b 5.5 c .02)/ out=resid outresid;
  run;
  ods html close;
```

The ODS HTML statement specifies three files. The BODY= option specifies the file to contain the output generated from the statements that follow. The BODY= option is the only required option.

The CONTENTS= option specifies a file to contain the table of contents. The FRAME= option specifies a file to contain both the table of contents and the output. You open the FRAME= file in your browser to view the table of contents together with the generated output (see Output 6.2.1). Note that, if you specify the ODS HTML statement with only the BODY= argument, no table of contents is created.

The MODEL procedure is invoked to fit the specified model. The resulting output is displayed in Output 6.2.1.
Output 6.2.1. HTML Output from the MODEL Procedure

The table of contents displayed in Output 6.2.1 contains the descriptive label for each output table produced in the MODEL procedure step. You can select any label in the table of contents and the corresponding output will be displayed in the right-hand side of the browser window.
Example 6.3. Determining the Names of ODS Tables

In order to select or exclude a table, or to render it as a SAS data set, you must first know its name. You can obtain the table names in several ways:

- For any SAS/ETS procedure, you can obtain table names from the individual procedure chapter or from the SAS online Help system.
- For any SAS procedure, you can use the SAS Explorer window to view the names of the tables created in your SAS run.
- For any SAS procedure, you can use the ODS TRACE statement to find the names of tables created in your SAS run. The ODS TRACE statement writes identifying information to the SAS log for each generated output table.

This example uses the ODS TRACE statement with the LISTING option to obtain the names of the created output objects. By default, the ODS TRACE statement writes its information to the SAS log. However, you can specify the LISTING option to have the information interleaved with the procedure output in the SAS listing.

The model will be the U.S. population model from the previous example.

```sas
ods trace on/listing;

title 'Logistic Growth Curve Model of U.S. Population';
proc model data=uspop;
    label a = 'Maximum Population'
            b = 'Location Parameter'
            c = 'Initial Growth Rate';
    pop = a / ( 1 + exp( b - c * (year-1790) ) );
    fit pop start=(a 1000 b 5.5 c .02)/ out=resid outresid;
run;

ods trace off;
```

The purpose of these statements is to obtain the names of the ODS tables produced in this PROC MODEL run. The ODS TRACE ON statement writes the trace record of ODS output tables. The LISTING option specifies that the information is interleaved with the output and written to the SAS listing.

The MODEL procedure is invoked to perform the analysis, the SAS listing receives the procedure output and the trace record, and the trace is then turned off with the OFF option.
Output 6.3.1. The ODS Trace, Interleaved with MODEL Results: Partial Results

The MODEL Procedure

Output Added:
-------------
Name: ResidSummary
Label: Nonlinear OLS Summary of Residual Errors
Template: ets.model.ResidSummary
Path: Model.OLS.ResidSummary
-------------

Nonlinear OLS Summary of Residual Errors

<table>
<thead>
<tr>
<th>Equation</th>
<th>DF</th>
<th>DF Adj</th>
<th>SSE</th>
<th>MSE</th>
<th>R-Square</th>
<th>R-Sq</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>pop</td>
<td>3</td>
<td>18</td>
<td>345.6</td>
<td>19.2020</td>
<td>0.9972</td>
<td>0.9969</td>
<td>U.S. Population in Millions</td>
</tr>
</tbody>
</table>

Output Added:
-------------
Name: ParameterEstimates
Label: Nonlinear OLS Parameter Estimates
Template: ets.model.ParameterEstimates
Path: Model.OLS.ParameterEstimates
-------------

Nonlinear OLS Parameter Estimates

<table>
<thead>
<tr>
<th>Parameter</th>
<th>Estimate</th>
<th>Approx Std Err</th>
<th>t Value</th>
<th>Pr &gt;</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>a</td>
<td>387.9307</td>
<td>30.0404</td>
<td>12.91</td>
<td>&lt;.0001</td>
<td>Maximum Population</td>
</tr>
<tr>
<td>b</td>
<td>3.990385</td>
<td>0.0695</td>
<td>57.44</td>
<td>&lt;.0001</td>
<td>Location Parameter</td>
</tr>
<tr>
<td>c</td>
<td>0.022703</td>
<td>0.00107</td>
<td>21.22</td>
<td>&lt;.0001</td>
<td>Initial Growth Rate</td>
</tr>
</tbody>
</table>

As displayed in Output 6.3.1, the ODS TRACE ON statement writes the name, label, template, and path name of each generated ODS table. For more information on names, labels, and qualified path names, see the discussion in the section “Using the Output Delivery System” beginning on page 182.

The information obtained with the ODS TRACE ON statement enables you to request output tables by name. The examples that follow demonstrate how you can use this information to select, exclude, or create data sets from particular output tables.
Example 6.4. Selecting ODS Tables for Display

You can use the ODS SELECT statement to deliver only certain tables to open ODS destinations. In the following example, the MODEL procedure is used to fit a model for new one-family home sales.

```sas
title 'Modeling One-Family Home Sales';
data homes;
  input year q pop yn cpi @@;
y=yn/cpi;
  label q='New One-Family Houses Sold in Thousands'
    pop='U.S. Population in Millions'
    y='Real Personal Income in Billions'
    cpi='U.S. CPI 1982-1984 = 100';
cards;
  70 485 205.052 715.6 .388 71 656 207.661 776.8 .405
  72 718 209.896 839.6 .418 73 634 211.909 949.8 .444
  74 519 213.854 1038.4 .493 75 549 215.973 1142.8 .538
  76 646 218.035 1252.6 .569 77 819 220.239 1379.3 .606
  78 817 222.585 1551.2 .652 79 709 225.055 1729.3 .726
  80 545 227.719 1918.0 .824 81 436 234.296 2428.1 .996
  82 412 232.171 2261.4 .965 83 623 234.296 2283.7 1.076
  84 639 236.343 2668.6 1.039 85 688 238.466 2838.7 1.136
  86 750 240.658 3013.3 1.096 87 671 242.820 3194.7 1.136
  88 676 245.051 3479.2 1.183 89 650 247.350 3725.5 1.240
  90 536 249.975 3945.8 1.307
;
ods select ResidSummary ParameterEstimates;
ods trace on;
ods show;

The ODS SELECT statement specifies that only the two tables “ResidSummary” and “ParameterEstimates” are to be delivered to the ODS destinations. In this example, no ODS destinations are explicitly opened. Therefore, only the SAS listing, which is open by default, receives the procedure output. The ODS SHOW statement displays the current overall selection list in the SAS log. The ODS TRACE statement writes the trace record of the ODS output objects to the SAS log. In the following statements, the MODEL procedure is invoked to produce the output.

```sas
proc model data=homes;
  parms a b c d;
  q = a + b*y + c*lag(y) + d*pop;
  %ar(ar_q,1,q)
  endo q;
  exo y pop;
  id year;
  fit q / dw;
run;
```

Output 6.4.1 displays the results of the ODS SHOW statement, which writes the cur-
rent overall selection list to the SAS log. As specified in the preceding ODS SELECT statement, only the two ODS tables “ResidSummary” and “ParameterEstimates” are selected for output.

**Output 6.4.1.** Results of the ODS SHOW Statement

```sas
ods select ResidSummary ParameterEstimates;
ods trace on;
ods show;
```

Current OVERALL select list is:
1. ResidSummary
2. ParameterEstimates

Partial results of the ODS TRACE statement, which is written to the SAS log, are displayed in Output 6.4.2.

**Output 6.4.2.** The ODS TRACE: Partial Contents of the SAS Log

```sas
proc model data=homes;
  parms a b c d;
  q = a + b*y + c*lag(y) + d*pop;
  %ar(ar_q,1,q)
  endo q;
  exo y pop;
  id year;
  fit q / dw;
run;
```

Output Added:
-------------
Name: ModSummary
Label: Variable Counts
Template: ets.model.ModSummary
Path: Model.ModSum.ModSummary
-------------

Output Added:
-------------
Name: ResidSummary
Label: Nonlinear OLS Summary of Residual Errors
Template: ets.model.ResidSummary
Path: Model.OLS.ResidSummary
-------------

Output Added:
-------------
Name: ParameterEstimates
Label: Nonlinear OLS Parameter Estimates
Template: ets.model.ParameterEstimates
Path: Model.OLS.ParameterEstimates
-------------

In the following statements, the ODS SHOW statement writes the current overall selection list to the SAS log. The QUIT statement ends the MODEL procedure. The second ODS SHOW statement writes the selection list to the log after PROC MODEL terminates. The ODS selection list is reset to ’ALL,’ by default, when a procedure terminates. For more information on ODS exclusion and selection lists, see the section “Using the Output Delivery System” beginning on page 182.
ods show;
quit;
ods show;

The results of the statements are displayed in Output 6.4.3. Before the MODEL
procedure terminates, the ODS selection list includes only the two tables, “Resid-
Summary” and “ParameterEstimates.”

Output 6.4.3.  The ODS Selection List, Before and After PROC MODEL Terminates

        ods show;
   Current OVERALL select list is:
    1. ResidSummary
    2. ParameterEstimates

        quit;

        NOTE: PROCEDURE MODEL used:
             real time    0.34 seconds
             cpu time     0.19 seconds

        ods show;
   Current OVERALL select list is: ALL

The MODEL procedure supports run-group processing. Before the QUIT statement
is executed, PROC MODEL is active and the ODS selection list remains at its pre-
vious setting before PROC MODEL was invoked. After the QUIT statement, the
selection list is reset to deliver all output tables.

The entire displayed output consists of the two selected tables, as displayed in Out-
put 6.4.4.
### Logistic Growth Curve Model of U.S. Population

The MODEL Procedure

#### Nonlinear OLS Summary of Residual Errors

<table>
<thead>
<tr>
<th>Equation</th>
<th>DF Model</th>
<th>DF Error</th>
<th>SSE</th>
<th>MSE</th>
<th>R-Square</th>
<th>Adj R-Sq</th>
<th>Durbin Watson</th>
</tr>
</thead>
<tbody>
<tr>
<td>q</td>
<td>5</td>
<td>15</td>
<td>86388.2</td>
<td>5759.2</td>
<td>0.6201</td>
<td>0.5188</td>
<td>1.7410</td>
</tr>
</tbody>
</table>

#### Nonlinear OLS Parameter Estimates

| Parameter | Estimate  | Approx Std Err | t Value | Pr > |t| | Label        |
|-----------|-----------|----------------|---------|------|----|----------------|
| a         | 2622.538  | 1196.5         | 2.19    | 0.0446|
| b         | 1.216858  | 0.3723         | 3.27    | 0.0052|
| c         | -0.65809  | 0.3676         | -1.79   | 0.0936|
| d         | -14.8418  | 8.6435         | -1.72   | 0.1065|
| ar_q_l1   | 0.478075  | 0.2480         | 1.93    | 0.0730| AR(ar_q) q lag1 parameter |
Example 6.5. Creating an Output Data Set from an ODS Table

The ODS OUTPUT statement creates SAS data sets from ODS tables. In the following example, the AUTOREG procedure is invoked to estimate a large number of Dickey-Fuller type regressions and part of the resulting procedure output is output to a SAS data set. The Dickey-Fuller t-statistic is then calculated and PROC MEANS is used to calculate the empirical critical values.

The data set UNITROOT contains 10,000 unit root time series.

```
data unitroot;
  YLag = 0;
  do rep = 1 to 10000;
    do time = -50 to 100;
      Y = YLag + rannor(123);
      if time > 0 then output;
      YLag = Y;
    end;
  end;
run;
```

Determining the Names of the ODS Tables

The purpose of the following statements is to obtain the names of the output tables produced in this PROC AUTOREG run. Note that a smaller data set, test, is used for this trial run. The ODS TRACE statement lists the trace record, and the SAS listing destination is closed so that no output is displayed.

```
data test;
  YLag = 0;
  do time = -50 to 100;
    Y = YLag + rannor(123);
    if time > 0 then output;
    YLag = Y;
  end;
run;
ods trace on;
ods listing close;
proc autoreg data=test;
  model Y = YLag;
run;
ods trace off;
ods listing;
```
Output 6.5.1. The ODS TRACE: Partial Contents of the SAS Log

```sas
ods trace on;
ods listing close;
proc autoreg data=test;
  model Y = YLag;
run;
```

Output Added:

---
Name: Dependent
Label: Dependent Variable
Template: ets.autoreg.Dependent
Path: Autoreg.Model1.Dependent
---

Output Added:

---
Name: ParameterEstimates
Label: Parameter Estimates
Template: ets.autoreg.ParameterEstimates
Path: Autoreg.Model1.OLSEst.ParameterEstimates
---

By default, the trace record is written to the SAS log, as displayed in Output 6.5.1. Note that you can alternatively specify that the information be interleaved with the procedure output in the SAS listing (see Example 6.3).

Creating the Output Data Set

In the statements that follow, the ODS OUTPUT statement writes the ODS table “ParameterEstimates” to a SAS data set called myParms. All of the usual data set options, such as the KEEP= or WHERE= options, can be used in the ODS OUTPUT statement. Thus, to modify the ParameterEstimates data set so that it contains only certain variables, you can use the data set options as follows.

```sas
ods output ParameterEstimates = myParms
  (keep=Variable Estimate StdErr
   where=(Variable='YLag')) ;
```

```sas
ods exclude all;
proc autoreg data=unitRoot;
  by rep;
  model Y = YLag;
run;
```

The KEEP= option in the ODS OUTPUT statement specifies that only the variables `Variable`, `Estimate`, and `StdErr` are written to the data set. The WHERE= option selects the specific variable in which we are interested, `YLag`. The AUTOREG procedure is again invoked. In order to limit the amount of displayed output, the ODS exclusion list is set to ALL.

In the following statements, the output data set `myParms` is used to create the data set `TDISTN` which contains the Dickey-Fuller t-statistics. PROC MEANS is then utilized.
to tabulate the empirical 1, 5, and 10 percent critical values. The results are displayed in Output 6.5.2.

```
data tdistn;
  set myParms;
  tStat = (Estimate-1)/StdErr;
run;

ods select Means.Summary;
proc means data=tDistn P1 P5 P10 fw=5;
  var tStat;
  title 'Simulated Dickey-Fuller Critical Values';
run;
```

**Output 6.5.2.** The Empirical Critical Values, Tabulated by PROC MEANS

<table>
<thead>
<tr>
<th>Analysis Variable : tStat</th>
</tr>
</thead>
<tbody>
<tr>
<td>1st</td>
</tr>
<tr>
<td>Pctl</td>
</tr>
<tr>
<td>------</td>
</tr>
<tr>
<td>-3.51</td>
</tr>
<tr>
<td>------</td>
</tr>
</tbody>
</table>

Simulated Dickey-Fuller Critical Values